```
' ====================================================================
'   W A T E R M A R K   M A C R O   ( Version 6.0a)
'   Based on Watermark macro suppiled by Microsoft WinWord
'   Modified by Daniel Kiechle (CIS 74640,120), October 1993
'   Übersetzt von Hagen Fuhrmann (CIS 100010,3362), Mai 1994
'   The following changes have been implemented:
'     1. The macro now works with any unit of measurement
'     2. The watermark is positioned correctly when in landscape mode
'     3. Improved default angle of rotation is now supplied
'     4. Waternark centering algorithm has been improved
'     5. A frame may now be printed around the page(s)
'     6. The user can now choose one of three print intensities
'     7. Existing watermarks are removed before new ones are added
'     8. Vertical/Horizontal alignment options were removed.
'     9. The watermark dialog has been redesigned.
' ====================================================================

Sub MAIN
Print "Wasserzeichen-Makro (Version 6.0a)"

If AuswInfo(27) Then
        MsgBox "Das Kommando kann im Makrofenster nicht benutzt werden"
        Goto bye
End If

CR$ = Chr$(13) : Pi = 3.1415926
ID$ = "%Eingefügt vom Wasserzeichen-Makro (Version 6.0)" + CR$

' SAVE CURRENT WORD SETTINGS (restored at end)
fViewFieldCodes = AnsichtFeldfunktionen()
fViewOutline = AnsichtGliederung()
fViewPage = AnsichtLayout()
fViewZentraldok = AnsichtZentraldokument()
AnsichtNormal

' GET PAGE SIZE AND ORIENTATION. We don't care about the measurement
' units yet, but we are interested in the pages's height/width ratio)
Dim dlg As DateiSeiteEinrichten
GetCurValues dlg
uw$ = dlg.SeitenBreite
uwidth = Val(uw$)
uh$ = dlg.SeitenLänge
uheight = Val(uh$)
uorientation$ = Mid$("Hochformat Querformat ",(dlg.HochQuer * 11) + 1, 11)

' DETERMINE DEFAULT ANGLE OF ROTATION. WordBasic doesn't know trig
' functions, so we use the series arctan x = x - x^3/3 + x^5/5 - x^7/7 ... The series
' only converges if x < 1, so we need to calculate the angle 90-x in cases
' where uratio > 1 and readjust once the angle calculation has been done
' WordBasic does not support the ^ (power) operator either, so we need to
' multiply things out explicitly. The increasing powers of uratio are held
' in variable "power". The series gives the angle in radians; we thus need
' to multiply the result by 180 and divide by Pi to obtain degrees.

reciprocal = 0 : sign = - 1 : lowestPower = 3 : highestPower = 13
uratio = uheight / uwidth
If uratio > 1 Then uratio = 1 / uratio : reciprocal = 1
defrot = uratio : power = uratio

For i = lowestPower To highestPower Step 2
        power = power * uratio * uratio
```

```
        defrot = defrot + (sign * power / i)
        sign = sign * - 1
Next i

defrot = defrot * 180 / Pi
If reciprocal Then defrot  = 90 - defrot
defrot = Int(defrot + 0.5)

' USER DIALOG DEFINITION
Begin Dialog BenutzerDialog 493, 275, "Wasserzeichen-Parameter"
        Text 18, 4, 200, 13, "&Wasserzeichen-Text:"
        TextBox 18, 18, 320, 18, .usertext
        CheckBox 360, 19, 120, 16, "&Rahmen", .userframe
        GroupBox 18, 43, 164, 72, "Zeichensatz"
        OptionGroup .userfont
                OptionButton 34, 58, 130, 16, "&Courier"
                OptionButton 34, 76, 130, 16, "&Helvetica"
                OptionButton 34, 94, 130, 16, "&Times-Roman"
        GroupBox 200, 43, 274, 72, "Attribute"
                CheckBox 215, 64, 91, 16, "&Fett", .userbold
                CheckBox 215, 88, 120, 16, "&Großbuchst.", .usercap
                CheckBox 360, 64, 100, 16, "&Kursiv", .useritalic
                CheckBox 360, 88, 100, 16, "&Schattiert", .usershade
        Text 18, 126, 164, 13, "Größe (&Punkte):"
        TextBox 18, 140, 164, 18, .usersize
        Text 18, 165, 164, 13, "&Drehwinkel:"
        TextBox 18, 179, 164, 18, .userrotate
        GroupBox 200, 126, 130, 72, "&Setze auf "
        OptionGroup  .userpage
                OptionButton 215, 146, 110, 16, "&Alle Seiten"
                OptionButton 215, 170, 110, 16, "&Erste Seite"
        GroupBox 344, 126, 130, 72, "Intensität"
        OptionGroup .userintensity
                OptionButton 360, 141, 100, 16, "He&ller"
                OptionButton 360, 159, 100, 16, "Nor&mal"
                OptionButton 360, 177, 100, 16, "Dun&kler"
        Text 20, 205, 450, 13, "Seite: " + uh$ + " x " + uw$ + ";  Ausrichtung: " + uorientation$
        Text 20, 220, 450, 13, "Hinweis: Druck nur mit Postscriptdrucker."
        OKButton 100, 240, 130, 21
        CancelButton 264, 240, 130, 21
        Text 461, 261, 31, 13, "-dk-"
End Dialog

' SET DIALOG DEFAULT VALUES
Redim dlg As BenutzerDialog
dlg.usertext = "ENTWURF"
dlg.userframe = 0
dlg.userfont = 1
dlg.userbold = 1
dlg.useritalic = 0
dlg.usercap = 0
dlg.usershade = 0
dlg.usersize = "100"
dlg.userrotate =  Mid$(Str$(defrot), 2)
dlg.userpage = 0
dlg.userintensity = 1

' SHOW DIALOG
prompt: On Error Goto bye
Dialog dlg
Print "Thanks!"
```

```
On Error Goto 0

' GET DIALOG VALUES (after dialog dismissed)
uframe = dlg.userframe
ubold = dlg.userbold
uitalic = dlg.useritalic
ushade = dlg.usershade
usize = Int(Val(dlg.usersize))
urotate = Int(Val(dlg.userrotate))
upage = dlg.userpage
shade = (98 - (2 * dlg.userintensity) + (1 * dlg.usershade))
shade$ = Str$(Int(shade / 100)) + "."
shade$ = shade$ + LTrim$(Str$(Int(shade - Int(shade / 100) * 100)))
dark = 52 - (20 * dlg.userintensity)
dark$ = Str$(Int(dark / 100)) + "."
dark$ = dark$ + LTrim$(Str$(Int(dark - Int(dark / 100) * 100)))

If dlg.usercap Then
	utext$ = UCase$(dlg.usertext)
Else
	utext$ = dlg.usertext
End If

' SELECTION OF THE APPROPRIATE POSTSCRIPT FONT
If dlg.userfont = 0 Then
	If ubold Then
		If uitalic Then
			fontname$ = "/Courier-BoldOblique"
		Else
			fontname$ = "/Courier-Bold"
		End If
	ElseIf uitalic Then
		fontname$ = "/Courier-Oblique"
	Else
		fontname$ = "/Courier"
	End If
ElseIf dlg.userfont = 1 Then
	If ubold Then
		If uitalic Then
			fontname$ = "/Helvetica-BoldOblique"
		Else
			fontname$ = "/Helvetica-Bold"
		End If
	ElseIf uitalic Then
		fontname$ = "/Helvetica-Oblique"
	Else
		fontname$ = "/Helvetica"
	End If
ElseIf dlg.userfont = 2 Then
	If ubold Then
		If uitalic Then
			fontname$ = "/Times-BoldItalic"
		Else
			fontname$ = "/Times-Bold"
		End If
	ElseIf uitalic Then
		fontname$ = "/Times-Italic"
	Else
		fontname$ = "/Times-Roman"
	End If
End If
```

```
' PARAMETER ERROR HANDLING
' Set minimum and maximum font sizes and degrees of rotation values
minsize = 4
maxsize = 127
minrot = 0
maxrot = 360

' ERROR MESSAGE SETUP
msg0$ = "Die Zeichengröße muß zwischen " + Str$(minsize) + " und  " + Str$(maxsize) + " liegen. "
msg1$ = "Der Drehwinkel muß zwischen " + Str$(minrot) + " und " + Str$(maxrot) + " Grad liegen."

fgotoprompt = 0
If usize < minsize Or usize > maxsize Then
        MsgBox msg0$, "Font Size Error", 16
        fgotoprompt = 1
End If
If fgotoprompt <> 1 Then
        If urotate < minrot Or urotate > maxrot Then
                MsgBox msg1$, "Falscher Drehwinkel", 16
                fgotoprompt = 1
        End If
End If
If fgotoprompt Then Goto prompt

' DETERMINE CURRENT MEASUREMENT UNITS
' (0 = inches 1 = centimeters 2 = points 3 = picas)
' We will need the page height in points, so we convert it if necessary.
Redim dlg As ExtrasOptionenAllgemein
GetCurValues dlg
If dlg.Maße = 0 Then
        ' convert inches to points
        uheight = uheight  * 72
ElseIf dlg.Maße = 1 Then
        ' convert centimeters to points
        uheight = uheight  * 28.35
ElseIf dlg.Units = 3 Then
        ' convert picas to points
        uheight = uheight  * 12
End If

' SET UP THE POSTSCRIPT CODE STRINGS
' (PSx$, where  x=1: setup, x=2: frame, x=3: orientation, x=4-12: watermark)
' The strings hold the code which will be sent to the PostScript interpreter

PS1$ = "Druck \p page " + Chr$(34) + CR$ + ID$ + "initclip" + CR$
PS2$ = ".5 setlinewidth" + CR$ + "wp$box clippath stroke"  + CR$
PS3$ =  Str$(uheight) + " 0 translate" + CR$ + "90 rotate" + CR$
PS4$ = fontname$ + " findfont " + Str$(usize) + " scalefont setfont " + CR$
PS5$ = "/TextToCenter (" + utext$ + ") def" + CR$
PS6$ = "/printDraft {0 0 moveto TextToCenter show} def" + CR$
PS7$ = "wp$x 2 div wp$y 2 div translate" + CR$
PS8$ =  Str$(urotate) + " rotate" + CR$
PS9$ = "TextToCenter stringwidth pop 2 div neg" + CR$
PS10$ = Str$(usize) + " 2 mul 3 div 2 div neg translate" + CR$
PS11$ = shade$ + "  -.05 " + dark$ + CR$ + "{setgray printDraft -1 .5 translate} for" + CR$
PS12$ =  shade$ + " setgray printDraft " + Chr$(34)

If uframe = 0 Then PS2$ = ""
If uorientation$ = "Hochformat " Then PS3$ = ""
If ushade = 0 Then PS11$ = ""
```

```
MarkPS$ = PS3$ + PS4$ + PS5$ + PS6$ + PS7$ + PS8$ + PS9$ + PS10$ + PS11$ + PS12$
If utext$ = "" Then MarkPS$ = ""

PostScript$ = PS1$ + PS2$ + MarkPS$

' DETERMINE EXISTING HEADER TYPES
' fheader = 0: no special first page, no special odd/even
' fheader = 1: special first page, no special odd/even
' fheader = 2: no special first page, special odd/even
' fheader = 3: special first page, special odd/even
Redim dlg As AnsichtNormalKopfzeileBereich
GetCurValues dlg
ffirst = dlg.ErsteSeite
fodd = dlg.GeradeUngeradeSeiten
If fodd = 0 Then
        If ffirst = 0 Then
                fheader = 0
        Else
                fheader = 1
        End If
Else
        If ffirst = 0 Then
                fheader = 2
        Else
                fheader = 3
        End If
End If

' INSERT POSTSCRIPT CODE into the appropriate headers. If First Page only is
' specified and there is no current special first page header, the contents
' of the normal header must be copied into the new First Page header. At the
' same time, any existing watermark code should be removed from other headers.

AnsichtFeldfunktionen 1
If upage = 0 Then
' Print Watermark on all pages
        If fheader = 0 Then
        ' No special first page header, no special odd and even page headers
                AnsichtNormalKopfzeileBereich .Art = 0, .ErsteSeite = 0, .GeradeUngeradeSeiten = 0
                InsertPS ID$, PostScript$
        ElseIf fheader = 1 Then
        ' Special first page header, no separate odd and even page headers
                AnsichtNormalKopfzeileBereich .Art = 0, .ErsteSeite = 1, .GeradeUngeradeSeiten = 0
                InsertPS ID$, PostScript$
                AnsichtNormalKopfzeileBereich .Art = 0, .ErsteSeite = 1, .GeradeUngeradeSeiten = 0
                InsertPS ID$, PostScript$
        ElseIf fheader = 2 Then
        ' No special first page header, special odd and even page headers
                AnsichtNormalKopfzeileBereich .Art = 0, .ErsteSeite = 0, .GeradeUngeradeSeiten = 1
                InsertPS ID$, PostScript$
                AnsichtNormalKopfzeileBereich .Art = 2, .ErsteSeite = 0, .GeradeUngeradeSeiten = 1
                InsertPS ID$, PostScript$
        ElseIf fheader = 3 Then
        ' Special first page header, special odd even page headers
                AnsichtNormalKopfzeileBereich .Art = 0, .ErsteSeite = 1, .GeradeUngeradeSeiten = 1
                InsertPS ID$, PostScript$
                AnsichtNormalKopfzeileBereich .Art = 2, .ErsteSeite = 1, .GeradeUngeradeSeiten = 1
                InsertPS ID$, PostScript$
                AnsichtNormalKopfzeileBereich .Art = 4, .ErsteSeite = 1, .GeradeUngeradeSeiten = 1
                InsertPS ID$, PostScript$
        EndIf
```

```
ElseIf upage = 1 Then
' Watermark on first page only
        If fheader = 0 Then
                AnsichtNormalKopfzeileBereich .Art = 0, .ErsteSeite = 0, .GeradeUngeradeSeiten = 0
                BearbeitenAllesMarkieren
                BearbeitenKopieren
                AusschnittSchließen
                AnsichtNormalKopfzeileBereich .Art = 2, .ErsteSeite = 1, .GeradeUngeradeSeiten = 0
                BearbeitenEinfügen
                InsertPS ID$, PostScript$
        ElseIf fheader = 1 Then
                AnsichtNormalKopfzeileBereich .Art = 0, .ErsteSeite = 1, .GeradeUngeradeSeiten = 0
                RemovePS ID$
                AnsichtNormalKopfzeileBereich .Art = 2, .ErsteSeite = 1, .GeradeUngeradeSeiten = 0
                InsertPS ID$, PostScript$
        ElseIf fheader = 2 Then
                AnsichtNormalKopfzeileBereich .Art = 2, .ErsteSeite = 0, .GeradeUngeradeSeiten = 1
                RemovePS ID$
                BearbeitenAllesMarkieren
                BearbeitenKopieren
                AusschnittSchließen
                AnsichtNormalKopfzeileBereich .Art = 0, .ErsteSeite = 1, .GeradeUngeradeSeiten = 1
                BearbeitenEinfügen
                InsertPS ID$, PostScript$
        ElseIf fheader = 3 Then
                AnsichtNormalKopfzeileBereich .Art = 2, .ErsteSeite = 1, .GeradeUngeradeSeiten = 1
                RemovePS ID$
                AnsichtNormalKopfzeileBereich .Art = 4, .ErsteSeite = 1, .GeradeUngeradeSeiten = 1
                RemovePS ID$
                AnsichtNormalKopfzeileBereich .Art = 0, .ErsteSeite = 1, .GeradeUngeradeSeiten = 1
                InsertPS ID$, PostScript$
        EndIf
EndIf
AusschnittSchließen

' RESTORE WORD SETTINGS saved at beginning of macro execution
If fViewOutline Then AnsichtGliederung
If fViewPage Then AnsichtLayout
If fViewZentraldok Then AnsichtZentraldokument
AnsichtFeldfunktionen fViewFieldCodes
bye:
End Sub

' SUBROUTINE TO INSERT POSTSCRIPT STRINGS
' (Remove any existing watermark field first)
Sub InsertPS(ID$, PostScript$)
        BeginnDokument
        NächstesFeld
        If InStr(Markierung$(), ID$) Then BearbeitenAusschneiden
        BeginnDokument
        EinfügenFeldzeichen
        Einfügen PostScript$
End Sub

' SUBROUTINE TO REMOVE EXISTING WATERMARKS
' (This routine is used when the watermark is to appear on the first
' page only since other headers may already contain watermarks.)
Sub RemovePS(ID$)
        BeginnDokument
        NächstesFeld
        If InStr(Markierung$(), ID$) Then BearbeitenAusschneiden
```

```
End Sub
```